
*                                                 VCC   ^
*                            VCC  ^           R1       /_\
*                                /_\       _/\  /\  /\__|
*                                 |      |    \/  \/
*                               __|_____/|\_______
*                              |         |        |
*                       R2      ->QP1    |  QP2 <-        R3
*          __________/\  /\  /\___`|_____|____|'____/\  /\  /\_
*         |            \/  \/    _'|   VPB    |`_     \/  \/   |
*         |  VNC1               |                |             |
*         |   _________________/|\_______________|      _______|
*         |  |                  |     ^     ^    |     |
*         |  |                  |    /_\   /_\   |     |
*         |  |     _____________|     |     |    |     |
*         |  |    | VNC2        |   <-      ->   |     |
*         |  |    |             | ||          || |     |
*         |  |    |             |_||MP1    MP2||_|     |
*         |  |    |             | ||_        _|| |     |
*         |  |    |             |    |      |    |     |
*         |_/|\__/|\_______     ->   |      |   <-     |
*            |    |        |      || |VG3   | ||       |
*            |    |        |   MP3||_|      |_||MP4    |
*            |    |        |     _|| |   VG4| ||_      |
*           _|    |_       |    |   _|_    _|_   |     |
*       __|'QN1  QN2`|_    | VG5|  / _ \  / _ \  |VG6  |
*      |  |`->    <-'| |   |    |  \/ \/  \/ \/  |     |
*      |     |____|   _|_  ->   |  /\_/\  /\_/\  |   <-
*      |        |     ///    || |  \___/  \___/  | ||
*      |    VNE_|_       MP5 ||_|   _|_    _|_   |_||MP6
*      |      / _ \         _|| |   ///    ///   | ||_
*      |      \/ \/I1      |    | I2          I3 |    |
*      |      /\_/\        |    |_______||_______|    |
*      |      \___/        |   _|_      ||      _|_   |
*      |       _|_         |  / _ \     C0     / _ \  |
*      |       ///         |  \/ \/            \/ \/  |
*      |________________   |  /\_/\            /\_/\  |
*                       |  |  \___/            \___/  |
*                       |  |   _|_ I4       I5  _|_   |
*      ____             |  |   ///              ///   | OUT2
*     |   _|_           |  |                          |  ____
*     |  /VOS\INP |\    |  |_/\  /\  /\___/\  /\  /\__|_|    |  ___
*     |  \___/  __|+\   |  |   \/  \/  _|_  \/  \/      |    |_|INN|
* VIN |    |____| |  \__|  |  R4       ///      R5      |BDIF| |___|
*    _|_          |  /  |  |____________________________|    |   |
*   /_  \   INN __|-/   |                               |____|   |
*  // \  \     |  |/    |   ____                                 |
*  \   \//     |        |__|OUT0|                                |
*   \___/      | B100dB    |____|                                |
*    _|_       |_________________________________________________|
*    ///   



.OPTIONS    GMIN = 1e-18 
VCC         VCC   0    DC    5
I4          VG5   0    20u
I5          VG6   0    20u
MP5         OUT1  VG5  VS5   VCC  pchannel   w=10u l=1u
MP6         OUT2  VG6  VS6   VCC  pchannel   w=10u l=1u
QP1         VNC2  VPB  VCC        PNPL
QP2         VNC1  VPB  VCC        PNPL
R1          VCC   VPB  10K
R2          VPB   VS5  30K
R3          VPB   VS6  30K
MP1         VG3   VNC2 VCC   VCC  pchannel   w=10u l=1u
MP2         VG4   VNC1 VCC   VCC  pchannel   w=10u l=1u
MP3         VG5   VG3  VNC2  VCC  pchannel   w=10u l=1u
MP4         VG6   VG4  VNC1  VCC  pchannel   w=10u l=1u
I2          VG3   0    20u
I3          VG4   0    20u
R4          OUT1  0    10
R5          OUT2  0    10
C0          VG5   VG6  10p
BDIF        INN   0    V  =  V(OUT2) -V(OUT1)

QN1         VNC1  OUT0 VNE       NPNV
QN2         VNC2  0    VNE       NPNV
I1          VNE   0    .5n

VIN         VIN   0    SIN( 0 .1u 10k) AC .1u
VOS         VIN   INP  DC    1f
B100dB      OUT0   0    V  = (V(INP) -V(INN))*100000

.MODEL  PNPL    PNP( BF=50 VAF=20 )

.MODEL  NPNV    NPN( BF=50 VAF=20 )

.model  nchannel nmos (level=3)

.model  pchannel pmos (level=3)

.control
set pensize = 2
tran        1u   1m    0  1u
plot        v(inp)
plot        v(out0)
ac          dec      30       1M      100k
plot        db(inp)  db(out0) db(out0/inp) db(inn/out0)

.endc

.end

=====================END_OF_SPICE============================

The integrator block gets connected to the 
100dB gain block and now the required gain
can be done on chip.  



Testing the ability to reject offset appears to be
pushing the convergence limits in some versions
of spice. But the AC plots show the integrator 
block functioning as desired. 



*                                                 VCC   ^
*                            VCC  ^           R1       /_\
*                                /_\      __/\  /\  /\__|
*                                 |      |    \/  \/
*                               __|_____/|\_______
*                              |         |        |
*                       R2      ->QP1    |  QP2 <-        R3
*          __________/\  /\  /\___`|_____|____|'____/\  /\  /\_
*         |            \/  \/    _'|   VPB    |`_     \/  \/   |
*         |  VNC1               |                |             |
*         |   _________________/|\_______________|      _______|
*         |  |                  |     ^     ^    |     |
*         |  |                  |    /_\   /_\   |     |
*         |  |     _____________|     |     |    |     |
*         |  |    | VNC2        |   <-      ->   |     |
*         |  |    |             | ||          || |     |
*         |  |    |             |_||MP1    MP2||_|     |
*         |  |    |             | ||_        _|| |     |
*         |  |    |             |    |      |    |     |
*         |_/|\__/|\_______     ->   |      |   <-     |
*            |    |        |      || |VG3   | ||       |
*            |    |        |   MP3||_|      |_||MP4    |
*            |    |        |     _|| |   VG4| ||_      |
*           _|    |_       |    |   _|_    _|_   |     |
*       __|'QN1  QN2`|_    | VG5|  / _ \  / _ \  |VG6  |
*      |  |`->    <-'| |   |    |  \/ \/  \/ \/  |     |
*      |     |____|   _|_  ->   |  /\_/\  /\_/\  |   <-
*      |        |     ///    || |  \___/  \___/  | ||
*      |    VNE_|_       MP5 ||_|   _|_    _|_   |_||MP6
*      |      / _ \         _|| |   ///    ///   | ||_
*      |      \/ \/I1      |    | I2          I3 |    |
*      |      /\_/\        |    |_______||_______|    |
*      |      \___/        |   _|_      ||      _|_   |
*      |       _|_         |  / _ \     C0     / _ \  |
*      |       ///         |  \/ \/            \/ \/  |
*      |________________   |  /\_/\            /\_/\  |
*                       |  |  \___/            \___/  |
*                       |  |   _|_ I4       I5  _|_   |
*      ____             |  |   ///              ///   | OUT2
*     |   _|_           |  |                          |  ____
*     |  /VOS\INP |\    |  |_/\  /\  /\___/\  /\  /\__|_|    |  ___
*     |  \___/  __|+\   |  |   \/  \/  _|_  \/  \/      |    |_|INN|
*     |    |____| |  \__|  |  R4       ///      R5      |BDIF| |___|
* VIN |           |  /  |  |____________________________|    |   |
*    _|_    INN __|-/   |                               |____|   |
*   /_  \      |  |/    |   ____                                 |
*  // \  \     |        |__|OUT0|                                |
*  \   \//     | B100dB    |____|                                |
*   \___/      |_________________________________________________|
*    _|_
*    ///



Input signal levels and gain is shown below. 

There were three integration blocks in the LMC2001.
One was used to gain up the chopper signal as shown
above. Another one was used to integrate the output
of the chopper signal. A third was used to actually
gain up the shot noise of an input stage which 
was then applied to spread spectrum the clock. 
As before, this was done in a way which did not 
introduce clock gliches on the supply. 

For some reason National Semiconductor applied for
three patents on the concept of actually using 
real random noise to spread spectrum the chopper. 
Perhaps the wording on the claims need some rework.


